

```

/*
Flash - Arduino Example script
version 1.3 : 12-05-2010
copyleft : Kasper Kamperman - Art & Technology - Saxion

More info on how to setup up Arduino Flash communication :
http://www.kasperkamperman.com/blog/arduino/arduino-flash-communication-as3/
The included 'readme.rtf'.

Summary :
- Set pinmodes (input, output, pwm, servo) in the defaultPinConfig array to the setup
you use.
- Change the speed of the timer to your preference (now 25 fps).
- Comment 'addChild(monitor)' to hide the monitor by default.
- keyHandler : 'Shift + m' shows/hides monitor object.
- Read/set inputs-outputs in the timerEvent function. The value of analog pin 0 is
now connected to the y-position of the ball instance on the stage.
*/

import net.eriksjodin.arduino.Arduino;
import net.eriksjodin.arduino.ArduinoWithServo;
import net.eriksjodin.arduino.events.ArduinoEvent;
import net.eriksjodin.arduino.events.ArduinoSysExEvent;

import com.kasperkamperman.monitor.ArduinoMonitor;

// == VARIABLES =====

// Change this array to the pin configuration you use in your own setup.

var defaultPinConfig:Array = new Array(
    null,          // Pin 0    null (is RX)
    null,          // Pin 1    null (is TX)
    'digitalIn',   // Pin 2    digitalIn or digitalOut
    'digitalIn',   // Pin 3    pwmOut or digitalIn or digitalOut
    'digitalIn',   // Pin 4    digitalIn or digitalOut
    'digitalIn',   // Pin 5    pwmOut or digitalIn or digitalOut
    'digitalIn',   // Pin 6    pwmOut or digitalIn or digitalOut
    'digitalIn',   // Pin 7    digitalIn or digitalOut
    'digitalIn',   // Pin 8    digitalIn or digitalOut
    'servo',       // Pin 9    pwmOut or digitalIn or digitalOut or servo
    'digitalIn',   // Pin 10   pwmOut or digitalIn or digitalOut or servo
    'digitalIn',   // Pin 11   pwmOut or digitalIn or digitalOut
    'digitalIn',   // Pin 12   digitalIn or digitalOut
    'digitalOut',  // Pin 13   digitalIn or digitalOut ( led connected )
    'analogIn',    // Analog pin 0  analogIn
    'analogIn',    // Analog pin 1  analogIn
    'analogIn',    // Analog pin 2  analogIn
    'analogIn',    // Analog pin 3  analogIn
    'analogIn',    // Analog pin 4  analogIn
    'analogIn'     // Analog pin 5  analogIn
);

var a:ArduinoWithServo;
var monitor:ArduinoMonitor;

// connect to a serial proxy on port 5331
a = new ArduinoWithServo("127.0.0.1", 5331);

// make a monitor object and connect it to the Arduino
monitor = new ArduinoMonitor(a, defaultPinConfig);

// listen for connection
a.addEventListener(Event.CONNECT,onSocketConnect);
a.addEventListener(IOErrorEvent.IO_ERROR,errorHandler);

// listen for firmware (sent on startup)
a.addEventListener(ArduinoEvent.FIRMWARE_VERSION, onReceiveFirmwareVersion);

// make a timer object that calls the timerEvent function 20 times a second (every 50ms)
var timer = new Timer(50);
timer.addEventListener("timer", timerEvent);

```

```

// == SETUP AND INITIALIZE CONNECTION ( don't modify ) =====

// triggered when there is an IO Error
function errorHandler(errorEvent:IOErrorEvent):void
{
    trace("- "+errorEvent.text);
    trace("- Did you start the Serproxy program ?");
}

// triggered when a serial socket connection has been established
function onSocketConnect(e:Object):void
{
    trace("- Connection with Serproxy established.");

    // request the firmware version
    a.requestFirmwareVersion();
}

function onReceiveFirmwareVersion(e:ArduinoEvent):void
{
    trace("- Connection with Arduino - Firmata version: " + String(e.value));

    trace("- Set default pin configuration.");

    // set Pinmodes by the default array.
    for(var i:int = 2; i<defaultPinConfig.length; i++)
    { // set digital output pins
        if(defaultPinConfig[i] == "digitalOut") a.setPinMode(i, Arduino.OUTPUT);
        // set digital input pins
        if(defaultPinConfig[i] == "digitalIn") a.setPinMode(i, Arduino.INPUT);
        // set pwm output pins
        if(defaultPinConfig[i] == "pwmOut") a.setPinMode(i, Arduino.PWM);
        // set servo output pins
        if(defaultPinConfig[i] == "servo")
        { a.setupServo(i, 0);
          // write set start position to 0 otherwise it turns directly to 90 degrees.
          a.writeAnalogPin(i, 0);
        }
    }

    // you have to turn on reporting for every ANALOG pin individually.
    for(var j:int = 0; j<6; j++)
    { a.setAnalogPinReporting(j, Arduino.ON);
    }

    // for digital pins its only one setting
    a.enableDigitalPinReporting();

    startProgram();
}

// == START PROGRAM AND MONITOR =====

function startProgram()
{
    // start the timer that calls the timerEvent function
    timer.start();

    // show/hide the monitor with a keypress
    stage.addEventListener(KeyboardEvent.KEY_DOWN, keyHandler, false, 0, true);

    // display the monitor object. Comment to hide on startup
    addChild(monitor);
}

```

```

// function to show/hide the Arduino monitor with Shift + M
function keyHandler(e:KeyboardEvent):void
{
    // keyCode 77 + shiftKey = M
    if (e.keyCode == 77 && e.shiftKey)
    {
        if(contains(monitor))
        {
            trace("- hide Arduino monitor");
            removeChild(monitor);
        }
        else
        {
            trace("- show Arduino monitor");
            addChild(monitor);
        }

        stage.focus=stage;
    }
}

// == YOUR PROGRAM HERE =====

/*

How the get data from the Arduino :

    a.getDigitalData(<pin number>);
    example : a.getDigitalData(2);

    a.getAnalogData(<analog pin number>);
    example : a.getAnalogData(0);

    note : you can only get data from a pin if its configured as INPUT
           input.

Set Arduino outputs :

    a.writeDigitalPin(<pin number>, <0 or 1>);
    example : a.writeDigitalPin(13, 1);

    When configured as PWM : a.writeAnalogPin(<pin number>, <0 - 255>);
    When configured as Servo : a.writeAnalogPin(<pin number>, <0 - 179>);

    example : a.writeAnalogPin(9,128);

    note : to write digital data the pin has to be configured as OUTPUT
           to write analog data the pin has to be configured as PWM
           to write servo position data the pin has to be configured as servo

*/

/* In the function timerEvent we change the y position of the movieclip instance 'ball'.
- The data from Analog input 0 is used ( value between 0 and 1023 ).
- The position is calculated with the input value.
- The movement is smoothed with a factor ( between 0 -1 )
*/

function timerEvent(event:Event):void
{
    // calculate position
    var position:Number;
    position = a.getAnalogData(0) * ((stage.stageHeight-ball.height)/1023);

    // smooth factor, between 0 - 1
    var factor:Number = 0.6;

    // set ball y position
    ball.y = (factor * ball.y) + ((1-factor) * position);
}

```